
Chapter 2
Lectures on Geometry of Monge–Ampère
Equations with Maple

Alexei Kushner, Valentin V. Lychagin and Jan Slovák

2.1 Introduction

Themain goal of these lectures is to give a brief introduction to application of contact
geometry to Monge–Ampère equations. These equations have the form

Avxx + 2Bvxy + Cvyy + D(vxx vyy − v2xy) + E = 0, (2.1)

where A, B, C, D and E are functions on independent variables x, y, unknown
function v = v(x, y) and its first derivatives vx , vy .

Equations of this type arise in various fields. For example, G. Monge considered
such equations in connection with the problem of the optimal transportation of sand
or soil. This problemwas of great importance for the construction of fortifications. A
modernmodification of this problemhas the applications tomathematical economics,
especially in taxations problem (Kantorovich–Monge problem [7]).

J.G. Darboux studied and applied such equations in his lectures on general theory
of surfaces [3–5]. At that time, geometry was a source of various types of equations.
For example, the problem of reconstructing a surfacewith a givenGaussian curvature
K (x, y) is equivalent to solving the following equation:

vxx vyy − v2xy = K (x, y)
(
1 + v2x + v2y

)2
. (2.2)

Nowadays, the number of sources of Monge–Ampère equations has increased.
Equations arise in physics, aerodynamics, hydrodynamics, filtration theory, inmodels
of the development of oil and gas fields, in meteorology and so on. Some of these
applicationswill be discussed.On the other hand, aswe shall see, theMonge–Ampère
equations themselves generate geometric structures. For instance, some hyperbolic
equations can be considered as almost product structures, and elliptic ones as almost
complex structures.

The class of equations is rather wide and contains all linear and quasi-linear
equations as we can see. On the other hand, it is the minimal class that contains
quasi-linear equations and that is closed with respect to contact transformations.

This fact was known to Sophus Lie, who applied contact geometry methods to
this kind of equations. In this paper, S. Lie posed some classification problems for
equations with respect to contact pseudogroup. In particular, he posed the problem
of equivalence of equations to the quasi-linear and linear forms. This problem was
solved byV.V. Lychagin and V.N. Rubtsov [20] (see also [21]) in symplectic case and
byA.G. Kushner [12] in contact case. Conditions when equations can be transformed
to equationswith constant coefficients by contact transformationswere found byD.V.
Tunitskii [23]. The problem of classification for mixed type equations was solved by
A.G. Kushner [9–11].

In 1978, V.V. Lychagin noted that the classical Monge–Ampère equations and
its multi-dimensional analogues admit effective description in terms of differential
forms on the space of 1-jets of smooth functions [16]. His idea was fruitful, and it
generated a new approach to Monge–Ampère equations.

The lectures has the following structure.
The first lecture is an introduction to geometry of 1-jets space. We define 1-jets

of scalar functions, Cartan distribution, contact transformations and contact vector
fields on the 1-jets space [8, 15].

In the second lecture, we describe V.V. Lychagin approach and an introduction
to geometry of the Monge–Ampère equations. We follow papers [16, 17] and books
[15, 18].

The third lecture is devoted to contact transformations of the Monge–Ampère
equations. We consider examples of such transformations and apply them to con-
struct multivalued solutions. We illustrate this on the example of equation arising
in filtration theory of two immiscible fluids (oil and water, for example) in porous
media [1].

In the fourth lecture, we study geometrical structures associated with non-
degenerated (i.e. hyperbolic and elliptic) equations. We consider also the class of
so-called symplectic equations and give a criterion of their linearization by symplec-
tic transformation [18, 19].



2 Lectures on Geometry of Monge–Ampère Equations with Maple 55

The last, fifth lecture is devoted to tensor invariants of the Monge–Ampère equa-
tions. We construct here differential 2-forms that generalize the well-known Laplace
invariants. We follow the papers [12, 14].

All calculations in these lectures are illustrated in the programMaple. The Maple
files can be found on the website d-omega.org.

2.2 Lecture 1. Introduction to Contact Geometry

2.2.1 Bundle of 1-Jets

Let M be an n-dimensional smoothmanifold,C∞(M) be the ring of smooth functions
on M and T ∗

a M be the cotangent space at the point a ∈ M .

Definition 2.1 A 1-jet [ f ]1a of a function f ∈ C∞(M) at the point a is a pair

( f (a), d f |a) ∈ R × T ∗M.

The set of 1-jets at the point a ∈ M of all functions

J 1
a M := {[ f ]1a | f ∈ C∞(M)

}

is a vector space with respect to operations of addition and multiplication by real
numbers which are pointwise is defined as

[ f ]1a + [g]1a := [ f + g]1a, k[ f ]1a := [k f ]1a .

Denote by
J 1M := R × T ∗M

the set of 1-jets of all smooth functions f ∈ C∞(M) at all points a ∈ M .
This is a smooth manifold of dimension 2 dim M + 1 with local coordinates

x1, . . . , xn u, p1, . . . , pn , where x1, . . . , xn are local coordinates on M , p1, . . . ,

pn are the induced coordinates on the cotangent bundle and u is the standard coor-
dinate on R. In other words, the values of these functions at point [ f ]1k ∈ J 1M are
the following:

xi ([ f ]1a) = xi (a), u([ f ]1a) = f (a), pi ([ f ]1a) = fxi (a), i = 1, . . . , n. (2.3)

These coordinates are called canonical.
In what follows we’ll call J 1M the manifold of 1-jets, and the projection

π1 : J 1M −→ M, where π1 : [ f ]1a �−→ a

the 1-jet bundle.
Any function f ∈ C∞(M) defines the following map:

j1( f ) : M −→ J 1M, (2.4)

where
j1( f ) : M � a �−→ [ f ]1a ∈ J 1

a M ⊂ J 1M.

The image
�1

f := j1( f )(M) ⊂ J 1M,

which is a smooth submanifold of J 1M , is called the 1-graph of the function f .
Consider the following differential 1-form

κ := du − p1dx1 − · · · − pndxn

on the 1-jet space J 1M which we’ll call Cartan form.
It is easy to check that this form does not depend on a choice of canonical coor-

dinates in J 1M .
This form allows us to separate submanifolds of the form �1

f ⊂ J 1M from arbi-
trary submanifolds of dimension n by observation that

κ|�1
f
= 0,

for any f ∈ C∞(M). Indeed,

κ|�1
f
= d f − fx1dx1 − · · · − fxi dxi = 0.

On the other hand, if a submanifold N ⊂ J 1M is a graph of section s : M −→
J 1M , i.e. π1 : N −→ M is a diffeomorphism, and

κ|N = 0,

then one can easily check that N = �1
f for some smooth function f ∈ C∞(M).

This observation shows that zeroes of the Cartan form (but not the form itself) is
important to distinguish 1-graphs from arbitrary submanifolds in J 1M .

Denote by C the 2n-dimensional distribution (Cartan distribution) on J 1M given
by zeroes of the Cartan form:

C : J 1M � θ �−→ C(θ) := kerκθ ⊂ Tθ (J 1M).

In the dual way, the Cartan distribution can be defined by vector fields tangent to
this distribution. Namely, vector fields

∂x1 + p1∂u, . . . , ∂xn + pn∂u, ∂p1 , . . . , ∂pn

give us a local basis in the module of vector fields tangent to C. This module will be
denoted by D(C).

Then a submanifold N ⊂ J 1M is a graph of a smooth function if and only if

1. N is an integral submanifold of the Cartan distribution and
2. π1 : N → M is a diffeomorphism.

Remind that a contact structure on an odd-dimensional manifold K , dim K =
2k + 1, consists of 2k-dimensional distribution P on K such that

λ ∧ (dλ)k 
= 0

for any differential 1-form λ, such that locally P = ker λ.
In our case, we have

κ ∧ (dκ)n 
= 0

and therefore the Cartan distribution defines the contact structure on the manifold of
1-jets J 1M .

2.2.2 Contact Transformations

A transformation � of the space J 1M is called contact, if it preserves the Cartan
distribution, i.e.

�∗(C) = C.

In terms of the Cartan form, a transformation � is contact if

�∗(κ) = h�κ (2.5)

for some function h�, or equivalently

�∗(κ) ∧ κ = 0.

Examples of Contact Transformations

1. Translations:

(x1, x2, u, p1, p2) �−→ (x1 + α1, x2 + α2, u + β, p1, p2),

where α1, α2 and β are constants.
2. The Legendre transformation:

(x1, x2, u, p1, p2) �−→ (p1, p2, u − x1 p1 − x2 p2,−x1,−x2).

3. Partial Legendre’s transformation:

(x1, x2, u, p1, p2) �−→ (p1, x2, u − p1x1,−x1, p2).

Infinitesimal versions of contact transformations are contact vector fields.
A vector field X on J 1M is called contact if its local translation group consists

of contact transformations.
It means that

�∗
t (κ) = λtκ (2.6)

for some function λt on J 1M . Here, �t are shifts along vector field X .
After differentiating both parts of (2.6) by t at t = 0, we get:

d

dt

∣∣
∣∣
t=0

(
�∗

t (κ)
) =

(
dλ

dt

∣∣
∣∣
t=0

)
κ.

The left-hand side of the equation is the Lie derivative L X (κ) of the Cartan form in
the direction of the vector field X and therefore, we get

L X (κ) = hκ,

where h is a function on J 1M .
Multiplying both parts of the last equation by κ, we get:

L X (κ) ∧ κ = 0. (2.7)

In canonical coordinates, each contact vector field has the form

X f = −
n∑

i=1

∂ f

∂pi

∂

∂xi
+

(

f −
n∑

i=1

pi
∂ f

∂pi

)
∂

∂u
+

n∑

i=1

(
∂ f

∂xi
+ pi

∂ f

∂u

)
∂

∂pi

for some function f which is called generating function of the contact vector field.
Note that

κ(X f ) = f.

Maple Code: Main Operation on J 1
R

2

1. Load libraries:

with(DifferentialGeometry): with(JetCalculus):

2. Set jet notation, declare coordinates on the manifold M and generate coordinates
on the 1-jet space:

Preferences("JetNotation", "JetNotation2"):

DGsetup( [x1,x2],[u], M, 1, verbose);

3. Generate the Cartan form:

kappa:= convert(Cu[0,0],DGform);

4. Define partial Legendre transformation:

PartLegendre:=Transformation(M,M,[x1=-u[1,0],x2=x2,

u[0,0]=u[0,0]-u[1,0]*x1, u[1,0]=x1, u[0,1]=u[0,1]]);

5. Apply this transformation to the Cartan form:

Pullback(PartLegendre,kappa);

6. Prolongation of transformations from J 0M to J 1M :

Phi:=Transformation(M,M,

[x1=x2,x2=x1+x2,u[0,0]=-u[0,0]]);

Prolong(Phi,1);

7. Define the contact vector field X f with generating function f = p2:

X:=GeneratingFunctionToContactVector(u[0, 1]);

8. Prolongation of vector fields from the plane M = R
2 to J 1M :

Y:=evalDG(-x2*D_x1+x_1*D_x2);

Prolong(Y,1);

2.3 Lecture 2. Geometrical Approach to Monge–Ampère
Equations

2.3.1 Non-linear Second-Order Differential Operators

Following [16], any differential n-form ω on J 1M is associated with the differential
operator

�ω : C∞(M) −→ �n(M),

which acts in the following way:

�ω(v) := j1(v)
∗(ω), (2.8)

where (see formula (2.4))

j1(v)
∗ : �n(J 1M) −→ �n(M).

This construction does not cover all non-linear second-order differential operators,
but only a certain subclass of them.

Examples

1. The differential 1-form on J 1
R

ω = (1 − x2)dp + (λu − xp) dx,

where

λ = a2

b2
,

generates the Lissajou differential operator

�ω(y) =
(

(1 − x2)y′′ − xy′ + a2

b2
y

)
dx . (2.9)

Indeed,

�ω(v) = (1 − x2)d
(
y′) +

(
−xy′ + a2

b2
y

)
dx

=
(

(1 − x2)y′′ − xy′ + a2

b2
y

)
dx .



2 Lectures on Geometry of Monge–Ampère Equations with Maple 61

2. The differential 2-form on J 1
R

2

ω = dp1 ∧ dp2

generates the Hesse operator

�ω(v) = (det Hess v) dx1 ∧ dx2. (2.10)

Indeed,

�ω(v) = d
(
vx1

) ∧ d
(
vx2

)

= (
vx1x1dx1 + vx1x2dx2

) ∧ (
vx2x1dx1 + vx2x2dx2

)

= (
vx1x1vx2x2 − v2x1x2

)
dx1 ∧ dx2

= (det Hess v) dx1 ∧ dx2,

where Hess v is the Hessian of the function v.
3. The differential 3-form

ω = p1dp1 ∧ dx2 ∧ dx3 − dx1 ∧ dp2 ∧ dx3 − dx1 ∧ dx2 ∧ dp3 (2.11)

on J 1
R

3 produces the von Karman differential operator

(
vx vxx − vyy − vzz

)
dx ∧ dy ∧ dz,

where x = x1, y = x2, z = x3.
4. The differential 2-form

ω = dp1 ∧ dx2 − dp2 ∧ dx1

on J 1
R

2 represents the two-dimensional Laplace operator

�ω(v) = (
vxx + vyy

)
dx ∧ dy,

where x = x1, y = x2.
5. Two differential 2-forms

ω = dx1 ∧ du and  = p2dx1 ∧ dx2 (2.12)

on J 1
R

2 generate the same operator:

�ω(v) = dx1 ∧ (
vx1dx1 + vx2dx2

) = vx2 dx1 ∧ dx2,
�(v) = vx2 dx1 ∧ dx2.

6. Any differential n-form
ω = κ ∧ α + dκ ∧ β (2.13)

on J 1M , where α ∈ �n−1
(
J 1M

)
, β ∈ �n−2

(
J 1M

)
and κ is the Cartan form,

gives the zero operator.

All differential operators �ω generate differential equations of second order:

�ω(v) = 0. (2.14)

For example, operator (2.9) generates Lissajou equation

(1 − x2)y′′ − xy′ + a2

b2
y = 0. (2.15)

Note that the differential operators �ω and �hω generate the same equation for
each non-zero function h.

Equation (2.14) are called Monge–Ampère equations [16].
The following observation justifies this definition: beingwritten in local canonical

contact coordinates on J 1M , the operators �ω have the same type of non-linearity
as the Monge–Ampère equations.

Namely, the non-linearity involves the determinant of the Hesse matrix and its
minors. For instance, in the case n = 2, for

ω =Edx1 ∧ dx2 + B (dx1 ∧ dp1 − dx2 ∧ dp2) + (2.16)

Cdx1 ∧ dp2 − Adx2 ∧ dp1 + Ddp1 ∧ dp2.

we get classical Monge–Ampère equations

Avxx + 2Bvxy + Cvyy + D(vxx vyy − v2xy) + E = 0. (2.17)

An advantage of this approach is the reduction of the order of the jet space: we use
the simpler space J 1M instead of the space J 2M where Monge–Ampère equations
should be ad hoc as second-order partial differential equations [8].

The differential equation which is associated with a differential n-form ω will be
denote by Eω:

Eω := {�ω(v) = 0}.

The following Maple code generates the corresponding differential operator �ω

for a differential 2-form ω on J 1
R

2.

Maple Code: ω �−→ �ω

with(DifferentialGeometry): with(JetCalculus):

Preferences("JetNotation", "JetNotation2"):

DGsetup( [x1,x2],[u], M, 1);

DGsetup( [x,y], N, verbose);

Construct the differential operator �:

Delta := proc(z, h)

Pullback(Prolong(Transformation(N,M,

[x1=x,x2=y,u[0,0]=h]),2),z);

end proc;

Define a differential 2-form:

omega:=evalDG(dx1 &w du[1,0]-dx2 &w du[0,1]);

Apply the differential operator to this differential formω = dx1 ∧ dp1 − dx2 ∧ dp2:

simplify(Delta(omega,v(x,y)),size);

As a result, we get the differential operator

2
∂2

∂y∂x
dx ∧ dy.

2.3.2 Multivalued Solutions of Monge–Ampère Equations

Let v be a classical solution of the Monge–Ampère equation Eω, i.e. �ω(v) = 0.
Then

j1(v)
∗(ω) = 0.

It means that the restriction of the differential form ω to 1-graph of the function v is
zero:

ω |�1
v
= 0.

An n-dimensional submanifold L ⊂ J 1M is called a multivalued solution of
Monge–Ampère equation if



64 A. Kushner et al.

1. L is an integral manifold of the Cartan distribution, i.e. the restriction of the
Cartan form to L is zero: κ |L= 0;

2. the restriction of the differential n-form ω to L is zero, too: ω |L= 0.

Examples: Multivalued Solutions

1. Parameterized curves

L =
{

x = sin bt, y = cos at, p = − a sin at

b cos bt

}

in the space J 1
R are multivalued solutions of the Lissajou equation

(1 − x2)y′′ − xy′ + a2

b2
y = 0. (2.18)

Indeed, the restriction of the differential 1-form

ω = (1 − x2)dp +
(

a2

b2
y − xp

)
dx

on the curve L is zero. The projections of these curves on the plane (x, y) are
well-known Lissajou curves (see Figs. 2.1, 2.2).

2. Projections of multivalued solutions of the Monge equation

vxx vyy − v2xy = (
1 + v2x + v2y

)2

to the space R
3 with coordinates x, y, v are spheres with radius 1 (see Eq. (2.2).

3. Projections of multivalued solutions of the equation

vxx vyy − v2xy = 0 (2.19)

to the space R
3 with coordinates x, y, v are deployable surfaces.

2.3.3 Effective Forms

Last two examples (2.12) and (2.13) show that the constructed map

“differential n-forms” → “differential operators”

has a huge kernel.

Fig. 2.1 Multivalued
solutions of the Lissajou
equation for a = 3, b = 2

Fig. 2.2 Multivalued
solutions of the Lissajou
equation for a = 1, b = √

2
is a curve, everywhere dense
in the square (Lissajou’s
Black Square)



66 A. Kushner et al.

This kernel consists of differential forms that vanish on any integral manifold of
the Cartan distribution. All such forms have form (2.13) (see [15]).

Let’s find a submodule of the module �2(J 1M) of differential 2-forms such that
the map is bijective (dim M = 2).

Differential 2-form ω ∈ �2(J 1M) is called effective if

1. X1ω = 0;
2. ω ∧ dκ = 0.

Here, X1 is the contact vector field with generating function 1. In canonical co-
ordinates (2.3)

X1 = ∂u .

The first condition means that coordinate representation of ω does not contain
terms du ∧ ∗, and therefore ω 
= κ ∧ α for some differential 1-form α. Second con-
dition means that ω 
= βdκ, for a function β.

The module of effective differential 2-forms will be denoted by �2
ε(J 1M).

There is the projection p which maps module �2(J 1M) to the module �2(C) of
“differential forms” on the Cartan distribution.

Namely, define
p : �2(J 1M) −→ �2(C)

as follows:
p(ω) := ω − κ ∧ (X1ω).

Here, �2(J 1M) and �2(C) are modules of 2-forms on the 1-jet manifold J 1M and
on the Cartan distribution C respectively. Remark that

X1p(ω) = 0,

i.e. 2-form p(ω) ∈ �2(C).

Theorem 2.1 Any differential 2-form ω ∈ �2(C) has the unique representation

ω = ωε + βdκ, (2.20)

where ωε ∈ �2
ε(J 1M) is an effective 2-form and β is a function.

Proof In our case, the Cartan distribution C is four-dimensional. The exterior dif-
ferential of the Cartan form is non-degenerated 2-form on each Cartan subspace, i.e.
dκθ is a symplectic structure on C(θ) for any θ ∈ J 1M . Therefore, formula

ω ∧ dκ = βdκ ∧ dκ

uniquely defines a function β. Define now differential form

ωε = ω − βdκ.

Since ωε ∧ dκ = 0, the form ωε is effective. �

The constructed differential form ωε is called the effective part of the differential
form ω.

Define the operator

Eff : �2(J 1M) −→ �2
ε(J 1M), Eff(ω) := (p(ω))ε,

which for any differential 2-form ω on the space J 1M gives its effective part.
It is obvious that differential 2-forms ω and Eff(ω) generate the same Monge–

Ampère equations.
In canonical coordinates

dκ = dx1 ∧ dp1 + dx2 ∧ dp2

and any effective differential 2-form has the following representation:

ω =Edx1 ∧ dx2 + B (dx1 ∧ dp1 − dx2 ∧ dp2) + (2.21)

Cdx1 ∧ dp2 − Adx2 ∧ dp1 + Ddp1 ∧ dp2,

where A, B, C, D and E are smooth functions on J 1M . This form corresponds to
Eq. (2.17).

The followingMaple code contains two procedures which generate effective parts
of differential 2-forms.

Maple Code: ω �−→ ωε

1. Projection of a 2-form to the Cartan distribution:

ProjC:=proc (omega)

GeneratingFunctionToContactVector(1);

evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),omega));

end proc:

2. Calculation of effective parts of a differential 2-forms:

Eff:=proc (omega)

evalDG(evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),

omega))-(solve(op(Tools:-DGinfo(evalDG(g*Omega&w Omega-

(evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),omega)))

&w Omega),"CoefficientSet")),g))*Omega);

end proc:

2.4 Lecture 3. Contact Transformations of Monge–Ampère
Equations

By the definition, contact transformations preserve the Cartan distribution and mul-
tiply the Cartan form κ by a function (see formula (2.5)).

Therefore, contact transformations do not preserve the contact vector field X1

in general. Because of this, the image of an effective differential form can be not
effective.

Let � : J 1M → J 1M be a contact transformation and ω be an effective differ-
ential 2-form. Then by the image of differential 2-form ω, we shall understand the
effective differential form Eff(�∗(ω)).

Two Monge–Ampère equations Eω and E are contact equivalent if there exist a
contact transformation � such that  = hEff(�∗(ω)) for some function h.

Theorem 2.2 If two equations Eω and E are contact equivalent, then their contact
transformation maps multivalued solutions of one to multivalued solutions of the
other.

Note that, in general, contact transformations do not preserve the class of classical
solutions: classical solutions can transform to multivalued solutions and vice versa.

Examples of Linearization of Equations by Contact Transformations

1. The von Karman equation
vx1vx1x1 − vx2x2 = 0 (2.22)

becomes the linear equation

x1vx2x2 + vx1x1 = 0 (2.23)

after Legendre transformation (2.24).
The last equation is known as the Triccomi equation.

2. Equation
det Hess v = 1

is generated by the effective differential 2-form

ω = dp1 ∧ dp2 − dx1 ∧ dx2.

After the partial Legendre transformation

� : (x1, x2, u, p1, p2) �→ (p1, x2, u − p1x1, −x1, p2)

this form becomes
ω = dx2 ∧ dp1 − dx1 ∧ dp2,

and corresponds to the Laplace equation

vx1x1 + vx2x2 = 0.

3. Quasi-linear equation:

A
(
vx , vy

)
vxx + 2B

(
vx , vy

)
vxy + C

(
vx , vy

)
vyy = 0.

This equation is represented by the following effective form:

ω =B (p1, p2) (dx1 ∧ dp1 − dx2 ∧ dp2) + C (p1, p2) dx1 ∧ dp2 − A (p1, p2) dx2 ∧ dp1.

After the Legendre transformation

� : (x1, x2, u, p1, p2) �→ (p1, p2, u − p1x1 − p2x2, −x1, −x2, ) (2.24)

we get the following effective form

ϕ∗(ω) = B (−x1,−x2) (dx1 ∧ dp1 − dx2 ∧ dp2)+
− A (−x1,−x2) dx1 ∧ dp2 + C (−x1,−x2) dx2 ∧ dp1,

which corresponds to the linear equation:

−A (−x1,−x2) vx2x2 + 2B (−x1,−x2) vx1x2 − C (−x1,−x2) vx1x1 = 0.

Example

The following equation arises in filtration theory of two immiscible fluids in porous
media [1]:

uxy − ux uyy = 0. (2.25)

It is used for finding a strategy to control wavefronts in the development of oil
fields.

The corresponding differential 2-form is

ω = 2p1dp2 ∧ dx1 + dx1 ∧ dp1 − dx2 ∧ dp2,

where x1 = x , x2 = y. Applying the Legendre transformation

� : (x1, x2, u, p1, p2) �−→ (p1, p2, u − x1 p1 − x2 p2,−x1,−x2)

we get the following differential 2-form:

�∗(ω) = 2x1dx2 ∧ dp1 + dx1 ∧ dp1 − dx2 ∧ dp2.

This form corresponds to the linear equation

ux1x2 − x1ux1x1 = 0. (2.26)

The general solution of the last equation is

u(x1, x2) = e−x2 F1(x1ex2) + F2(x2), (2.27)

where F1 and F2 are arbitrary functions. Differentiating both sides of (2.27), we get

ux1 = F ′
1(x1ex2),

ux2 = −e−x2 F1(x1ex2) − F ′
1(x1ex2)x1 + F ′

2(x2).

Thus, solution (2.27) generate a surface L ⊂ J 1M :

L :

⎧
⎪⎨

⎪⎩

u − e−x2 F1(x1ex2) + F2(x2) = 0,

p1 − F ′
1(x1ex2) = 0,

p2 + e−x2 F1(x1ex2) + F ′
1(x1ex2)x1 − F ′

2(x2) = 0.

Applying the inverse Legendre transformation

�−1 : (x1, x2, u, p1, p2) �−→ (−p1,−p2, u − x1 p1 − x2 p2, x1, x2)

to L , we get multivalued solutions of equation (2.25) in parametric form (Fig. 2.3):

�−1(L) :

⎧
⎪⎨

⎪⎩

u − x1 p1 − x2 p2 − ep2 F1(−p1e−p2) + F2(−p2) = 0,

x1 − F ′
1(−p1e−p2) = 0,

x2 + ep2 F1(−p1e−p2) + p1F ′
1(−p1e−p2) + F ′

2(−p2) = 0.

(2.28)

In order to simplify the last formula, we introduce new parameters

a = −p1e−p2 , b = −p2,

and new functions
k(a) = F1(a), r(b) = F2(b).

In these notation, multivalued solutions of equation (2.25) takes the form:

Fig. 2.3 Projection of the
multivalued solution L to the
space x, y, u for
k(a) = a9 − 20a5 and
r(b) = b0.01

L :

⎧
⎪⎪⎪⎪⎪⎪⎨

⎪⎪⎪⎪⎪⎪⎩

x = k ′(a),

y = e−b(ak ′(a) − k(a)) − r ′(b),

u = (b + 1)e−b(k(a) − ak ′(a)) + br ′(b) − r(b),

p1 = −ae−b,

p2 = −b,

where k(a) and r(b) are arbitrary functions.

Maple Code: Equation uxy − ux uyy = 0

Define coordinates on M :

with(DifferentialGeometry): with(JetCalculus):

Preferences("JetNotation", "JetNotation2"):

DGsetup( [x1,x2],[u], M, 1);

DGsetup( [x,y],N,1);

Construct the differential operator �:

Delta := proc(z, h)

description "M-A operator";

Pullback(Prolong(Transformation

(N,M,[x1=x,x2=y,u[0,0]=h]),2),z);

end proc;

Define the differential 2-form ω:

omega:=evalDG(2*u[1,0]*du[0,1] &w dx1 +

dx1 &w du[1,0]-dx2 &w du[0,1]);

ω = 2p1dp2 ∧ dx1 + dx1 ∧ dp1 − dx2 ∧ dp2,

The Legendre transformation:

Legendre:=Transformation(M,M,[x1=u[1,0],x2=u[0,1],

u[0,0]=u[0,0]-x1*u[1,0]-u[0,1]*x2, u[1,0]=-x1, u[0,1]=-x2]):

Apply the Legendre transformation to ω:

omega1:=Pullback(Legendre,omega);

Construct the differential operator �ω1 :

Delta(omega1,u(x,y));

(
2

(
∂2

∂y∂x
u(x, y)

)
− 2x

(
∂2

∂x2
u(x, y)

))
dx ∧ dy

Check solution:

sub:={u(x,y)=exp(-y)*F1(x*exp(y))+F2(y)};

eval(diff(u(x, y), x, y)-x*diff(u(x, y), x, x), sub);

0

Inverse Legendre transformation:

InvLegendre:=InverseTransformation(Legendre):

Apply this transformation to the surface L:

z1:=convert(u(x1,x2)-exp(-x2)*F1(x1*exp(x2))+F2(x2),DGjet):

z2:=convert(diff(u(x1,x2)-exp(-x2)*

F1(x1*exp(x2))+F2(x2),x1),DGjet):

z3:=convert(diff(u(x1,x2)-exp(-x2)*

F1(x1*exp(x2))+F2(x2),x2),DGjet):

u1:=Pullback(InvLegendre,z1):

u2:=Pullback(InvLegendre,z2):

u3:=Pullback(InvLegendre,z3):

As a result, we get formula (2.28).
Check that L is a multivalued solution of equation (2.25), i.e. ω | L = 0:

DGsetup( [x1,x2,u,p1,p2], M);

DGsetup( [a,b], N);

omega:=evalDG(2*p1*dp2 &w dx1 + dx1 &w dp1-dx2 &w dp2):

NtoM:=Transformation(N,M,[x1=diff(k(a),a),

x2=exp(-b)*(a*diff(k(a),a)-k(a))-diff(r(b),b),

u=(b+1)*exp(-b)*(-a*diff(k(a),a)+k(a))+b*diff(r(b),b)-r(b),

p1=-a*exp(-b), p2=-b]):

Pullback(NtoM,omega);

0

Visualization of the multivalued solution L:

plot3d(eval([diff(k(a),a),exp(-b)*(a*diff(k(a),a)-k(a))

-diff(r(b),b), (b+1)*exp(-b)*(-a*diff(k(a),a)+k(a))+

b*diff(r(b),b)-r(b)], {k(a)=aˆ9-20*aˆ5,r(b)=bˆ0.01}),

a = -1 .. 1, b = -6 .. 6);

2.5 Lecture 4. Geometrical Structures

2.5.1 Pfaffians

First of all, we remark that the restriction of the differential 2-form dκ on the Cartan
distribution

� = dκ |C
defines a symplectic structure on Cartan space C(θ) ⊂ Tθ (J 1M).

Using this structure and an effective 2-form ω ∈ �2
ε(J 1M) we define function

Pf(ω), called Pfaffian, in the following way [20]:

Pf(ω)� ∧ � = ω ∧ ω. (2.29)

This is a correct construction because ω ∧ ω and � ∧ � are 4-forms on the four-
dimensional Cartan distribution.

In the case when

ω =Edx1 ∧ dx2 + B (dx1 ∧ dp1 − dx2 ∧ dp2) + (2.30)

Cdx1 ∧ dp2 − Adx2 ∧ dp1 + Ddp1 ∧ dp2,

we get
Pf(ω) = B2 + DE − AC.

We say that the Monge–Ampère equation Eω is hyperbolic, elliptic or parabolic
at a domainD ⊂ J 1M if the function Pf(ω) is negative, positive or zero at each point
of D, respectively.

If the Pfaffian changes the sign in some points ofD, then the equation Eω is called
a mixed type equation (see [10]).

The hyperbolic and elliptic equations are called non-degenerate.

Maple Code: Pfaffian

kappa:=convert(Cu[0,0],DGform):

Omega:=ExteriorDerivative(kappa):

omega:=evalDG(dq1 &w du[1,0]+ du[0,0] &w du[0,1]):

Pf:=proc (omega)

solve(op(DGinfo(evalDG(z*Omega &w Omega-omega &w omega),

"CoefficientSet")),z)

end proc:

For example, the Pfaffian of the differential 2-form

ω = dx1 ∧ dp1 − dx2 ∧ dp2

which corresponds to wave equation uxy = 0 is equal to −1, and as we know this
equation is hyperbolic.

The Pfaffian of the differential 2-form

ω = dx1 ∧ dp2 − dx2 ∧ dp1



2 Lectures on Geometry of Monge–Ampère Equations with Maple 75

which corresponds to Laplace equation uxx + uyy = 0 is equal to 1. Indeed,

omega:=evalDG(dx1 &w du[1,0]-dx2 &w du[0,1]):

Pf(omega);

−1

omega:=evalDG(dx1 &w du[0,1]-dx2 &w du[1,0]):

Pf(omega);

1

2.5.2 Fields of Endomorphisms

The standard linear algebra allows us to construct a field of endomorphisms

Aω : D(C) −→ D(C)

which is associated with an effective 2-form ω. Here D(C) is the module of vector
fields tangent to C.

Namely, the 2-form � is non-degenerated on C and the operator Aω is uniquely
determined by the following formula [19]:

Aω X � = X ω (2.31)

for all vector fields X tangent to C.
Proposition 2.1 Operators Aω satisfy the following properties:

1. �(Aω X, X) = 0.
2. �(Aω X, Y ) = �(X, AωY ).

Proof 1. �(Aω X, X) = ω(X, X) = 0.
2. �(Aω X, Y ) = ω(X, Y ) = −ω(Y, X) = −�(AωY, X) = �(X, AωY ). �

Proposition 2.2 The squares of operators Aω are scalar and

A2
ω + Pf(ω) = 0. (2.32)

Proof First of all

Aω X(ω ∧ �) = (Aω Xω) ∧ � + ω ∧ (Aω X�).

Using Proposition2.1,

X(Aω X(ω ∧ �)) =ω(Aω X, X)� − (Aω Xω) ∧ (X�)

+ (Xω) ∧ (Aω X�) + �(Aω X, X)ω

= �(A2
ω X, X)� − (A2

ω X�) ∧ (X�)

+ (Aω X�) ∧ (Aω X�) + �(Aω X, X)ω

= −(A2
ω X�) ∧ (X�).

Since ω is effective, ω ∧ � = 0. Then

(A2
ω X�) ∧ (X�) = 0,

i.e. differential 1-forms A2
ω X� and X� are linearly dependent. Therefore the

square of the operator Aω is a scalar: A2
ω = α.

Let X ∈ D(C) be an arbitrary vector field. Applying the operators Aω X and X
to both parts of formula (2.29) we get

Pf(ω)(Aω X�) ∧ (X�) = (Aω Xω) ∧ (Xω) = (αX�) ∧ (Aω X�).

Then
(Pf(ω) + α)(Aω X�) ∧ (X�) = 0. (2.33)

Suppose that (Aω X�) ∧ (X�) = 0. Then the vector fields X and Aω X are
linearly dependent. Since X is an arbitrary vector field we see that the operator Aω

is scalar, i.e. Aω X = λX for any X . Then

Xω = Aω X� = λX�.

Thereforeω = λ�, which is impossible. So from (2.33), it follows that Pf(ω) + α =
0, i.e. A2

ω + Pf(ω) = 0. �

Let’s find a coordinate representation of the operator Aω. Let

∂

∂x1
+ p1

∂

∂u
,

∂

∂x2
+ p2

∂

∂u
,

∂

∂p1
,

∂

∂p2
(2.34)

be a local basis of the module D(C). Then formula (2.31) gives:

Aω =

∥∥∥∥∥∥
∥∥

B −A 0 −D
C −B D 0
0 E B C

−E 0 −A −B

∥∥∥∥∥∥
∥∥

(2.35)

in this basis.

Maple Code: Operator Aω

with(DifferentialGeometry): with(LinearAlgebra): with(Tensor):

Coordinates on the 1-jet space:

DGsetup( [x1,x2,u,p1,p2], J):

Cartan’s form and its exterior differential:

kappa:=evalDG(du-p1*dx1-p2*dx2):

Omega:=ExteriorDerivative(kappa):

Define 2-form ω:

omega:=evalDG(2*p1*dp2 &w dx1+ dx1 &w dp1-dx2 &w dp2);

Vector fields and 1-forms on Cartan’s distribution:

VectCartan:=evalDG([D_x1+p1*D_u,D_x2+p2*D_u,D_p1,D_p2]):

CovectCartan:=evalDG([dx1,dx2,dp1,dp2]):

Checking duality:

m := proc (i, j) options operator, arrow;

Hook(VectCartan[i],CovectCartan[j])

end proc:

Matrix(4,m):

Construct an arbitrary vector field on Cartan’s distribution:

V:=DGzip([a, b, c,d], VectCartan, "plus"):

V = a
∂

∂x1
+ b

∂

∂x2
+ (bp2 + ap1)

∂

∂u
+ c

∂

∂p1
+ d

∂

∂p2

General form of A = Aω. Here ai, j are arbitrary functions:

A:=evalDG(sum(sum(a[i,j]*VectCartan[i] &t

CovectCartan[j],i=1..4),j=1..4)):

Action of Aω on vector fields:

Act:=Z->convert(ContractIndices(evalDG(A &tensor Z),

[[2,3]]), DGvector):

Equations with respect to ai, j :

for i from 1 to 4 do

e[i]:=evalDG(Hook(Act(evalDG(VectCartan[i])),Omega)-

Hook(VectCartan[i], omega));

end do:

AEq:=[]:

for i from 1 by 1 to 4 do

AEq:=[op(AEq),op(GetComponents(e[i],CovectCartan))]

end do:

AEq;

− a3,1, −a4,1, −1 + a1,1, 2p1 + a2,1,

− a3,2, −a4,2, a1,2, 1 + a2,2,

1 − a3,3, −a4,3, a1,3, a2,3,

− 2p1 − a3,4, −1 − a4,4, a1,4, a2,4

sol:=solve(AEq,[a[1,1],a[1,2],a[1,3],a[1,4],

a[2,1],a[2,2],a[2,3],a[2,4],

a[3,1],a[3,2],a[3,3],a[3,4],

a[4,1],a[4,2],a[4,3],a[4,4]]);

assign(sol);

m := proc (i, j) options operator, arrow; a[i,j] end proc;

Am:=Matrix(4,4,m);

Aω =

∥∥∥∥∥
∥∥∥

1 0 0 0
−2p1 −1 0 0
0 0 1 −2p1

0 0 0 −1

∥∥∥∥∥
∥∥∥

(2.36)

Determinant(Am);

1

Am.Am;
∥∥
∥∥∥∥∥
∥

1 0 0 0
0 1 0 0
0 0 1 0
0 0 0 1

∥∥
∥∥∥∥∥
∥

2.5.3 Characteristic Distributions

Effective forms ω and hω, where h is any non-vanishing function, define the same
Monge–Ampère equation. Therefore, for a non-degenerated equation Eω the form ω

can be normed in such a way that |Pf(ω)| = 1. It is sufficient to replace ω by

ω√|Pf(ω)| . (2.37)

By (2.32), the hyperbolic equations generate a product structure

A2
ω,a = 1

and elliptic equations generate a complex structure

A2
ω,a = −1

on the Cartan space C(a) [18].
Therefore, a non-degenerated Monge–Ampère equation generates two two-

dimensional (complex—for elliptic case) distributions on J 1M , which are
eigenspaces of the operator Aω.

These distributions C+(a) and C−(a) correspond to the eigenvalues 1 and −1
for the hyperbolic equations or to ι and −ι for the elliptic ones, respectively. Here
ι = √−1.

The distributions C+ and C− are called characteristic.
The characteristic distributions are real for the hyperbolic equations and complex

for the elliptic ones. They are complex conjugate for the elliptic equations.

Proposition 2.3 ([18]) 1. The characteristic distributions C+ and C− are skew
orthogonal with respect to the symplectic structure �, i.e. �(X+, X−) = 0 for
X± ∈ D(C±).

2. On each of them, the 2-form � is non-degenerate.

On the other hand, any pair of arbitrary real distributions C1,0 and C0,1 on J 1M
such that

1. dimC1,0 = dimC0,1 = 2;
2. C = C1,0 ⊕ C0,1;
3. C1,0 and C0,1 are skew-orthogonal with respect to the symplectic structure �

determines the operator A. Therefore, a hyperbolic Monge–Ampère equation can be
regarded as such pair

{C1,0,C0,1
}
of distributions.

Maple Code: Characteristic Distributions

Calculation of eigenvalues and eigenvectors of the operator Aω:

EV,e:=Eigenvectors(Am):

Find the vector fields from the Cartan distribution

Cp:=[]:Cm:=[]:

for i from 1 to 4 do

if EV[i]=EV[1] then Cp:=[op(Cp),

(convert((Transpose(e[1..-1,i])),list))]

else

Cm:=[op(Cm),(convert((Transpose(e[1..-1,i])),list))]

end if

end do:

Vp1:=DGzip(Cp[1], VectCartan, "plus");

Vp2:=DGzip(Cp[2], VectCartan, "plus");

Vm1:=DGzip(Cm[1], VectCartan, "plus");

Vm2:=DGzip(Cm[2], VectCartan, "plus");

For example, the characteristic distribution C+ and C− of operator (2.36) are
generated by the following vector fields:

C+ =
〈

p1
∂

∂p1
+ ∂

∂p2
,

∂

∂x2
+ p2

∂

∂u

〉

and

C− =
〈

∂

∂p1
, p1

∂

∂x2
+ p1(p2 − 1)

∂

∂u
− ∂

∂x1

〉
.

2.5.4 Symplectic Monge–Ampère Equations

Monge–Ampère equation (2.17) is called symplectic if its coefficients A, B, C, D, E
do not depend on v.

In this case, the structures described above (effective differential forms, the dif-
ferential operator �ω, field of endomorphisms Aω) can be considered on the four-
dimensional cotangent bundle T ∗M instead of the five-dimensional jet bundle J 1M .

Below, we repeat main constructions for the symplectic case.
A smooth function f ∈ C∞(M) defines a section s f : M −→ T ∗M of the cotan-

gent bundle
π : T ∗M −→ M

by the following formula:
s f : a �−→ d fa .

Let ω be a differential 2-form on T ∗M . Define a differential operator

�ω : C∞(M) −→ �2(M), �ω(v) := (sv)
∗(ω).

Then equation �ω(v) = 0 is a symplectic Monge–Ampère equation.
Let� be the symplectic structure on T ∗M . In canonical coordinates x1, x2, p1, p2

on T ∗M
� = dx1 ∧ dp1 + dx2 ∧ dp2.

The differential form ω is said to be effective if

ω ∧ � = 0.

Pfaffian Pf(ω) of the differential 2-form ω is defined by the following equality:

Pf(ω)� ∧ � = ω ∧ ω,

and formula
Aω X � = X ω

defines the field of endomorphisms Aω on T ∗M .
The square of operator Aω is scalar:

A2
ω + Pf(ω) = 0.

Consider now the casewhen equation is non-degenerated, i.e. Pf(ω) 
= 0 on T ∗M .
Then, the operator Aω can be normed (see formula (2.37).

For hyperbolic equations we get almost product structure: A2
ω = 1, and for elliptic

ones we get almost complex structure: A2
ω = −1.

We say that two symplectic equation Eω and E are symplectically equivalent if
there exist a symplectic transformation � such that

�∗(ω) = h

for some function h.
The following theorem gives a criterion of symplectic equivalence of non-

degenerated Monge–Ampère equation to linear equations with constant coefficients.

Theorem 2.3 ([19]) Non-degenerated symplectic Monge–Ampère equation Eω is
symplectically equivalent to wave equation

vxx − vyy = 0 (2.38)

(in hyperbolic case), or to Laplace equation

vxx + vyy = 0

(in elliptic case) if and only if the Nijenhuis tensor

NAω
= 0, (2.39)

where Aω is the normed operator.

Recall that the Nijenhuis tensor NA of an operator A is a tensor field of rank (1,
2) given by

NA(X, Y ) := −A2[X, Y ] + A[AX, Y ] + A[X, AY ] − [AX, AY ]

for vector fields X and Y .
Condition (2.39) can be written in the following equivalent form [20]:

dω = 1

2
d (ln |Pf(ω)|) ∧ ω.

Maple Code: Symplectic Equation and Nijenhuis Tensor

Below we construct the operator Aω for non-linear wave equation

vxy = f (x, y, vx , vy). (2.40)

Then we calculate the Nijenhuis tensor NAω
and find conditions under which is

this equation symplectically equivalent to the linear wave equation with constant
coefficients.

with(DifferentialGeometry): with(Tools):

with(PDETools): with(Tensor):with(LinearAlgebra):

DGsetup( [x1,x2,p1,p2], M):

Omega:=evalDG(dx1 &w dp1+dx2 &w dp2):

omega:=evalDG(-2*f(x1,x2,p1,p2)*dx1 &w dx2+

dx1 &w dp1-dx2 &w dp2);

Vect:=evalDG([D_x1,D_x2,D_p1,D_p2]):

Covect:=evalDG([dx1,dx2,dp1,dp2]):

V:=DGzip([a, b, c,d], Vect, "plus"):

A:=evalDG(sum(sum(a[i,j]*(Vect[i] &t

Covect[j]),i=1..4),j=1..4)):

Act:=Z->convert(ContractIndices

(evalDG(A &tensor Z),[[2,3]]),DGvector):

for i from 1 to 4 do

e[i]:=evalDG(Hook(Act(evalDG(Vect[i])),Omega)-

Hook(Vect[i],omega));

end do:

AEq:=[]:

for i from 1 by 1 to 4 do AEq:=

[op(AEq),op(GetComponents(e[i], Covect))] end do:

sol:=solve(AEq,[a[1,1],a[1,2],a[1,3],a[1,4],

a[2,1],a[2,2],a[2,3],a[2,4],

a[3,1],a[3,2],a[3,3],a[3,4],

a[4,1],a[4,2],a[4,3],a[4,4]]);

assign(sol):

A:=DGsimplify(convert(A, DGtensor)):

N := TensorBrackets(A, A, "Frolicher--Nijenhuis"):

eq:=Tools:-DGinfo(N, "CoefficientSet");

pdsolve(eq);

As a result we get
f = F1(x1, x2),

where F is an arbitrary function.
So, Eq. (2.40) is symplectically equivalent to wave equation (2.38) if and only if

f is a function in x1 and x2 only.

Fig. 2.4 Splitting of the
tangent space Ta(J 1M)

2.5.5 Splitting of Tangent Spaces

Let us return to the space J 1M .
A non-degenerate equation is called regular if the derivatives C(k)

± (k = 1, 2, 3)
of the characteristic distributions are constant rank distributions, too.

Below we consider regular equations only. Then, the first derivatives of the char-
acteristic distributions

C(1)
± := C± + [C±,C±]

are three-dimensional. Their intersection

l := C(1)
+ ∩ C(1)

−

is a one-dimensional distribution, which is transversal to Cartan distribution.
Therefore, for hyperbolic equations, the tangent space Ta(J 1M) splits into the

direct sum (see Fig. 2.4)

Ta(J 1M) = C+(a) ⊕ l(a) ⊕ C−(a) (2.41)

at each point a ∈ J 1M [18].
For elliptic equations, we get a similar decomposition of the complexification of

Ta(J 1M). In this case, the distribution l is real, too.

2.6 Lecture 5. Tensor Invariants of Monge–Ampère
Equations

2.6.1 Decomposition of de Rham Complex

Let us construct the decomposition of the de Rham complex, which is generated by
the splitting of tangent spaces.

Decomposition (2.41) generates a decomposition of the module of exterior s-
forms (or its complexification for elliptic equations). Denote the distributions C+, l,
and C− by P1, P2, and P3, respectively.

Let D(J 1M) be the module of vector fields on J 1M , and let D j be the module
of vector fields tangent to distribution Pj .

Define the following submodules of modules of differential s-forms �s(J 1M):

�s
i := {α ∈ �s(J 1M)| Xα = 0 ∀ X ∈ D j , j 
= i} (i = 1, 2, 3).

Then we get the following decomposition of the module of differential s-forms
on J 1M :

�s(J 1M) =
⊕

|k|=s

�k, (2.42)

where k =(k1, k2, k3) is a multi-index, ki ∈ {0, 1, . . . , dim Pi },

|k| = k1 + k2 + k3,

and

�k :=
⎧
⎨

⎩

∑

j1+ j2+ j3=|k|
α j1 ∧ α j2 ∧ α j3 , where α ji ∈ �

ki
i

⎫
⎬

⎭
⊂

3⊗

i=1

�
ki
i .

Three first terms of the decomposition are presented in the diagram (see Fig. 2.5).
The exterior differential also splits into the direct sum

d =
⊕

|t|=1

dt,

where
dt : �k → �k+t.

Theorem 2.4 ([12]) If the multi-index t contains one negative component and this
component is −1, then the operator dt is a C∞(J 1M)-homomorphism, i.e.,

dt( f α) = f dtα (2.43)

Fig. 2.5 Decomposition of de Rham complex

for any function f and any differential form α ∈ �k.

Due to this theorem, we have the seven homomorphisms, and three of them are
zeroes. The non-trivial homomorphisms are the following:

d2,−1,0, d0,−1,2, d−1,1,1 and d1,1,−1.

2.6.2 Tensor Invariants

Consider a case
t = 1 j + 1k − 1s .

Then the differential dt is a C∞(J 1M)-homomorphism. Note that

d1 j +1k−1s : �1q → 0,

if q 
= s. Then, the only non-trivial of d1 j +1k−1s is the restriction to the module �1s :

d1 j +1k−1s : �1s → �1 j ∧ �1k .

Therefore, the homomorphism d1 j +1k−1s defines a tensor field of the type (2,1).
This tensor field we denote by τ1 j +1k−1s :

τ1 j +1k−1s ∈ �1 j ∧ �1k ⊗ Ds .

A unique non-trivial component of this tensor field is its restriction to �1s . Note
that

τ1 j +1k−1s : �1s → �1 j ∧ �1k

coincides with d1 j +1k−1s .
Tensor fields τ1 j +1k−1s are differential invariants of Monge–Ampère equations.

So, we get four tensors of (2,1)-type [12]:

τ2,−1,0, τ0,−1,2, τ−1,1,1 and τ1,1,−1. (2.44)

Maple Code: Tensor Invariants

Below, we present a program for calculating the tensor τ−1,1,1. The remaining tensors
can be found similarly after a small adjustment of the program. In this program, we
omit the calculation of the characteristic distributions. They must be calculated in
advance (see “Maple Code: Operator Aω” and “Maple Code: Characteristic distri-
butions”).

with(DifferentialGeometry): with(LinearAlgebra):

with(Tensor):with(Tools): with(PDETools):

DGsetup( [x1,x2,u,p1,p2], J):

kappa:=evalDG(du-p1*dx1-p2*dx2):

Omega:=ExteriorDerivative(kappa):

omega:=evalDG(2*u*dx2 &w dp1+ dx1 &w dp1-

dx2 &w dp2-2*k*p1ˆ2*dx1 &w dx2):

Construct the distribution l (transversal to the Cartan distribution).We are looking
for l as an intersection of derivatives of the characteristic distributions C (1)

− and C (1)
+ .

This intersection is one-dimensional and it is generated by the vector field Z which
we are looking for.

S:=evalDG(a1*Vp1+a2*Vp2+a3*LieBracket(Vp1,Vp2)-

(b1*Vm1+b2*Vm2+b3*LieBracket(Vm1,Vm2))):

sol:=solve(Tools:-DGinfo(S, "CoefficientSet"),

[a1,a2,a3,b1,b2,b3]):

assign(sol):

Z:=evalDG(a1*Vp1+a2*Vp2+a3*LieBracket(Vp1,Vp2)):

Basis of the module of vector fields on J 1M and dual basis:

BV:=[Vm1,Vm2,Vp1,Vp2,Z]:

BC:=evalDG(DualBasis(BV)):

Decomposition of de Rham complex. Bases of �1(J 1M) and �2(J 1M):

Lambda[1,0,0]:=evalDG([BC[1], BC[2]]);

Lambda[0,1,0]:=evalDG([BC[5]]);

Lambda[0,0,1]:=evalDG([BC[3], BC[4]]);

Lambda[2,0,0]:=evalDG([BC[1] &w BC[2]]); #1

Lambda[1,1,0]:=evalDG([BC[1] &w BC[5], BC[2] &w BC[5]]); #2,3

Lambda[1,0,1]:=evalDG([BC[1] &w BC[3], BC[1] &w BC[4],

BC[2] &w BC[3], BC[2] &w BC[4]]); #4,5,6,7

Lambda[0,1,1]:=evalDG([BC[3] &w BC[5], BC[4] &w BC[5]]); #8,9

Lambda[0,0,2]:=evalDG([BC[3] &w BC[4]]); #10

List of elements of the basis of �2:

Lambda2:=[op(Lambda[2,0,0]),op(Lambda[1,1,0]),

op(Lambda[1,0,1]), op(Lambda[0,1,1]),op(Lambda[0,0,2])];

Construct the tensor τ−1,1,1:

unassign(’z1’,’z2’,’z3’,’z4’,’z5’,’z6’,’z7’,’z8’,’z9’,’z10’);

Arbitrary differential 2-form:

V:=evalDG(DGzip([z1,z2,z3,z4,z5,z6,z7,z8,z9,z10],

Lambda2, "plus")):

Arbitrary 2-form from �1,0,0:

S:=evalDG(ExteriorDerivative(C1*Lambda[1,0,0][1]+

C2*Lambda[1,0,0][2])-V):

S_coeff:=Tools:-DGinfo(S, "CoefficientSet"):

sol:=solve(S_coeff,{z1,z2,z3,z4,z5,z6,z7,z8,z9,z10});

assign(sol);

Projection of a differential 2-form to �0,1,1:

Pr_011:=evalDG(DGzip([z8,z9],

[Lambda2[8],Lambda2[9]],"plus")):

Pr_011:=convert(Pr_011, DGtensor):

unassign(’a’,’b’,’c’,’d’):

Tau:=evalDG(a*Lambda[0,1,1][1] &t BV[1]+

b*Lambda[0,1,1][2] &t BV[1]+

c*Lambda[0,1,1][1] &t BV[2]+

d*Lambda[0,1,1][2] &t BV[2]):

aTau:=ContractIndices(evalDG(Tau &t

(C1*Lambda[1,0,0][1]+C2*Lambda[1,0,0][2])),[[3,4]]):

eq0:=DGsimplify(evalDG(aTau-Pr_011)):

eq:=Tools:-DGinfo(eq0, "CoefficientSet"):

e1:=op(eval(eq,{C1=1,C2=0})):

e2:=op(eval(eq,{C1=0,C2=1})):

sol:=solve([e1,e2],[a,b,c,d]):

assign(sol):

Tau1:=DGsimplify(Tau):

tau[-1, 1, 1]:=DGsimplify(Tau);

Example: Hunter–Saxton Equation

Consider the Hunter–Saxton equation

vtx = vvxx + κv2x , (2.45)

where κ is a constant. This equation is hyperbolic, and it has applications in the
theory of liquid crystals [6].

The corresponding effective differential 2-form and the operator Aω are the fol-
lowing:

ω = 2udq2 ∧ dp1 + dq1 ∧ dp1 − dq2 ∧ dp2 − 2κp2
1dq1 ∧ dq2

and

Aω =

∥∥∥∥
∥∥∥∥

1 2u 0 0
0 −1 0 0
0 −2κp2

1 1 0
2κp2

1 0 2u −1

∥∥∥∥
∥∥∥∥

.

Let’s take the following base in the module of vector fields on J 1M :

X1 = ∂

∂q1
+ p1

∂

∂u
+ κp2

1
∂

∂p2
,

X2 = ∂

∂p1
+ u

∂

∂p2
,

Z = ∂

∂u
+ (2 κ − 1) p1

∂

∂p2
,

Y1 = ∂

∂q2
+ κp2

1
∂

∂p1
− u

∂

∂q1
+ (p2 − up1)

∂

∂u
,

Y2 = ∂

∂p2
.

The dual basis of the module of differential 1-forms is

α1 = dq1 + udq2,

α2 = dp1 − κp2
1dq2,

θ = du − p1dq1 − p2dq2,

β1 = dq2,

β2 = dp2+ (1 − 2κ) p1du+ (κ−1) p2
1dq1+ (2κ − 1) p1 p2dq2 − udp1.

The vector fields X1, X2 and Y1, Y2 form bases in the modules D(C+) and D(C−)

respectively. Tensor invariants of Eq. (2.45) have the form

τ−1,1,1 = − (p1dq1 ∧ dq2 + dq2 ∧ du) ⊗
(

∂

∂q1
+ p1

∂

∂u
+ κp21

∂

∂p2

)
,

τ1,1,−1 = 2( κ − 1)
(
κp31dq1 ∧ dq2 + κp21dq2 ∧ du −

dp1 ∧ du − p1dq1 ∧ dp1 − p2dq2 ∧ dp1) ⊗ ∂

∂p2
,

τ2,−1,0 =
(

dq1 ∧ dp1 − κp21dq1 ∧ dq2 + udq2 ∧ dp1
)

⊗
(

∂

∂u
+ (2 κ − 1) p1

∂

∂p2

)
,

τ0,−1,2 =
(

dq2 ∧ dp2 + (1 − 2κ) p1dq2 ∧ du + (1 − κ) p21dq1 ∧ dq2 − udq2 ∧ dp1
)

⊗
(

∂

∂u
+ (2 κ − 1) p1

∂

∂p2

)
.

2.6.3 The Laplace Forms

Define bracket 〈α ⊗ X, β ⊗ Y 〉 for decomposable tensors α ⊗ X and β ⊗ Y of types
(2,1) as follows [12]:

〈α ⊗ X, β ⊗ Y 〉 = (Y α) ∧ (Xβ) .

For non-decomposable tensors the bracket is defined by linearity.
Define two differential 2-forms λ− and λ+ from the module �1,0,1 as “wedge

contractions” of the tensor fields:

λ+ := 〈
τ0,−1,2, τ1,1,−1

〉
, λ− := 〈

τ2,−1,0, τ−1,1,1
〉
. (2.46)

Then tensors (2.46) are called Laplace forms of Monge–Ampère equations Eω.

Example: Laplace Form for Linear Equations

For linear hyperbolic equation

vxy = a(x, y)vx + b(x, y)vy + c(x, y)v + g(x, y), (2.47)

the Laplace forms are

λ− = kdx ∧ dy and λ+ = −hdx ∧ dy, (2.48)

where
k = ab + c − by h = ab + c − ax (2.49)

are the classical Laplace invariants. This observation justifies our definition.
For linear elliptic equations

vxx + vyy = a(x, y)vx + b(x, y)vy + c(x, y)v + g(x, y), (2.50)

Laplace forms generalize Cotton invariants [2].

We emphasize that the classical Laplace invariants (2.49) of Eq. (2.50) are not
absolute invariants even with respect to transformations

φ : (x, y, v) �→ (X (x), Y (y), A(x, y)v), A(x, y) 
= 0 (2.51)

in contrast to forms λ±, which are contact invariants.

Example: Laplace Forms for Hunter–Saxton Equation

The Laplace forms for the Hunter–Saxton equation (2.45) are

λ− = −dq2 ∧ dp1, λ+ = 2 (1 − κ) dq2 ∧ dp1.

2.6.4 Contact Linearization of the Monge–Ampère Equations

It is well known that if the classical Lagrange invariants h and k of a linear hyperbolic
equation is zero, then the equation can be reduced to the wave equation (see [22],
for example).

Similar statement is true for the Monge–Ampère equations [14]:

Theorem 2.5 A hyperbolic Monge–Ampère equation is locally contact equivalent
to the wave equation

vxy = 0

if and only if its Laplace invariants are zero: λ+ = λ− = 0.

Corollary 2.1 The equation

vxy = f
(
x, y, v, vx , vy

)

is locally contact equivalent to the wave equation vxy = 0 if and only if the function
f has the following form:

f = ϕyvx + ϕx vy + (ϕv + �v)vx vy + R,

where the function R = R(x, y, v) satisfies to the following ordinary linear differ-
ential equation:

Rv = (ϕv + �v)R + ϕxy − ϕxϕy .

Solving this equation we get

R = eϕ+�

(∫
(ϕxy − ϕxϕy)e

−ϕ−�dv + g

)
,

where ϕ = ϕ(x, y, v), � = �(v), and g = g(x, y) are arbitrary functions.

The general problem of linearization of non-degenerated Monge–Ampère equa-
tions with respect to the contact transformations was solved in [13].

